Four DOF robot

Figure 1 is a schematic representation of a robot [Kane], with four links 1, 2, 3, and 4. The mass center of the link i is designated C_i, $i = 1, 2, 3, 4$. The dimensions $AC_2 = L_1$, $AC_1 = L_2$ and $C_3C_4 = L_3$ are shown in Fig. 1.

Link 1 rotates in a "fixed" newtonian reference frame (0) about a vertical axis fixed in both (0) and 1. The reference frame (0) has the unit vectors $\boldsymbol{\imath}_0$, $\boldsymbol{\jmath}_0$, $\mathbf{k}_0$ as shown in Fig. 1. The reference frame (1) of unit vectors $[\boldsymbol{\imath}_1, \boldsymbol{\jmath}_1, \mathbf{k}_1]$ is attached to link 1. The vertical unit vectors $\boldsymbol{\jmath}_0$ and $\boldsymbol{\jmath}_1$ are fixed in both (0) and 1. The first generalized coordinate q_1 denotes the radian measure of the angle between the axes of (0) and (1).

Link 1 supports link 2, and link 2 rotates relative to 1 about a horizontal axis fixed in both 1 and 2. The reference frame (2) of unit vectors $[\boldsymbol{\imath}_2, \boldsymbol{\jmath}_2, \mathbf{k}_2]$ is fixed in 2. The horizontal unit vectors $\boldsymbol{\imath}_1$ and $\boldsymbol{\imath}_2$ are fixed in both 1 and 2. The mass center C_2 is a point fixed in both 1 and 2. The second generalized coordinate q_2 denotes the radian measure of the angle between the axes of (1) and (2).

The link 2 supports link 3, and link 3 has a translational motions relative to 2. The generalized coordinate q_4 is the distance between the mass centers, C_2 and C_3, of 2 and 3, respectively.

The link 3 supports link 4, and link 4 rotates relative to 3 about an axis fixed in both 3 and 4. The reference frame (4) of unit vectors $[\boldsymbol{\imath}_4, \boldsymbol{\jmath}_4, \mathbf{k}_4]$ is fixed in 4. The unit vectors $\boldsymbol{\jmath}_2$ and $\boldsymbol{\jmath}_4$ are fixed in both 3 and 4. The mass center C_4 is a point fixed in both 3 and 4. The generalized coordinate q_3 is the radian measure of the rotation angle between 3 and 4.

The reference frame (4), $[\boldsymbol{\imath}_4, \boldsymbol{\jmath}_4, \mathbf{k}_4]$ is fixed in 4 in such a way that $\boldsymbol{\imath}_0 = \boldsymbol{\imath}_4$, $\boldsymbol{\jmath}_0 = \boldsymbol{\jmath}_4$, $\mathbf{k}_4 = \mathbf{k}_0$ when $q_1 = q_2 = q_3 = 0$.

The generalized speeds, u_1, u_2, u_3, u_4, are associated with the motion of a system, and can be introduced as

$$\begin{aligned}
u_1 &= \boldsymbol{\omega}_{40} \cdot \boldsymbol{\imath}_4. \\
u_2 &= \boldsymbol{\omega}_{40} \cdot \boldsymbol{\jmath}_4, \\
u_3 &= \boldsymbol{\omega}_{40} \cdot \mathbf{k}_4, \\
u_4 &= \dot{q}_4, \qquad (1)
\end{aligned}$$

where $\boldsymbol{\omega}_{40}$ denotes the angular velocity of 4 in (0). 0ne may verify that

$$\boldsymbol{\omega}_{40} = (\dot{q}_1 s_2 s_3 + \dot{q}_2 c_3)\boldsymbol{\imath}_4 + (\dot{q}_1 c_2 + \dot{q}_3)\boldsymbol{\jmath}_4 + (-\dot{q}_1 s_2 c_3 + \dot{q}_2 s_3)\mathbf{k}_4, \qquad (2)$$

where s_i and c_i denote $\sin q_i$ and $\cos q_i$, $i = 1, 2, 3$, respectively. Substitution into Eq. (1) then yields

$$\begin{aligned} u_1 &= \dot{q}_1 s_2 s_3 + \dot{q}_2 c_3, \\ u_2 &= \dot{q}_1 c_2 + \dot{q}_3, \\ u_3 &= -\dot{q}_1 s_2 c_3 + \dot{q}_2 s_3, \\ u_4 &= \dot{q}_4. \end{aligned} \tag{3}$$

Equation (3) can be solved uniquely for $\dot{q}_1, \dot{q}_2, \dot{q}_3, \dot{q}_4$. Specifically,

$$\begin{aligned} \dot{q}_1 &= (u_1 s_3 - u_3 c_3)/s_2, \\ \dot{q}_2 &= (u_1 c_3 + u_3 s_3), \\ \dot{q}_3 &= u_2 + (u_3 c_3 - u_1 s_3) c_2 / s_2, \\ \dot{q}_4 &= u_4. \end{aligned} \tag{4}$$

with singularities at $q_2 = 0°$ and $q_2 = 180°$ degrees posing no problem. Thus, u_r as defined in Eq. (1) form a set of generalized speeds for the robot. The mass of link r is m_r, $r = 1, 2, 3, 4$. The central inertia dyadic of link r, can be expressed as

$$\bar{I}_r = (I_{xr} \mathbf{\imath}_r) \mathbf{\imath}_r + (I_{yr} \mathbf{\jmath}_r) \mathbf{\jmath}_r + (I_{zr} \mathbf{k}_r) \mathbf{k}_r, \tag{5}$$

where I_{xr}, I_{yr}, I_{zr} are the central principal moments of inertia of $r = 1, 2, 3, 4$.

In the case of the robot, there are two kinds of forces that contribute to the generalized active forces F_r, $r = 1, 2, 3, 4$ namely, contact forces applied in order to drive links 1, 2, 3, 4, and gravitational forces exerted on the links by the Earth.

Considering, the contact forces, the set of such forces transmitted from newtonian frame (0) to link 1 (through bearings and by means of motor) is replaced with a couple of torque $\mathbf{T}_{01}$ together with a force $\mathbf{F}_{01}$ applied to 1 at C_1.

Similarly, the set of contact forces transmitted from 2 to 1 is replaced with a couple of torque $\mathbf{T}_{21}$ together with a force $\mathbf{F}_{21}$ applied to 1 at C_2 (which is a point fixed in 1). The law of action and reaction then guarantees that the set of contact forces transmitted from 1 to 2 is equivalent to a couple of torque -$\mathbf{T}_{21}$ together with the force - $\mathbf{F}_{21}$ applied to 2 at C_2.

Next, the set of contact forces exerted on 2 by 3 is replaced with a couple of torque $\mathbf{T}_{32}$ together with a force $\mathbf{F}_{32}$ applied to 2 at C_{32}, the point of

2 instantaneously coinciding with C_3. The set of forces exerted by 2 on 3 is, therefore equivalent to a couple torque -$\mathbf{T}_{32}$ together with the force -$\mathbf{F}_{32}$ applied to 3 at C_3.

Similarly, torque $\mathbf{T}_{43}$ and forces $\mathbf{F}_{43}$ come into play in connection with the interactions of 3 and 4 at C_4. As for gravitational forces exerted on the links of the robot by the Earth, these are denoted by $\mathbf{G}_r$, r = 1, 2, 3, 4 respectively.

The following notations are introduced

$$\begin{aligned}
\tau_1 &= \mathbf{T}_{01} \cdot \mathbf{j}_1 = k_1(q_{1f} - q_1) - k_2\dot{q}_1, \\
\tau_2 &= \mathbf{T}_{21} \cdot \mathbf{i}_2 = k_3(q_2 - q_{2f}) + k_4\dot{q}_2 + g[(m_3 + m_4)q_4 + m_4 L_3]s_2, \\
\tau_3 &= \mathbf{T}_{43} \cdot \mathbf{j}_3 = k_5(q_3 - q_{3f}) + k_6\dot{q}_3, \\
\sigma &= \mathbf{F}_{32} \cdot \mathbf{j}_3 = k_7(q_4 - q_{4f}) + k_8\dot{q}_4 - g(m_3 + m_4)c_2. \qquad (6)
\end{aligned}$$

where $k_1, ..., k_8$ are constant "gains" k_1 = 3.0 N m, k_2 = 5.0 N m s, k_3 = 1.0 N m, k_4 = 3.0 N m s, k_5 = 0.3 N m, k_6 = 0.6 N m s, k_7 = 30 N m, k_8 = 41 N s, and $q_{rf} = \pi/3$ rad $r = 1, 2, 3$ while q_{4f} = 0.1 m.

The initial numerical data are L_1 = 0.1 m, L_2 = 0.1 m, L_3 = 0.7 m, m_1 = 9 kg, m_2 = 6 kg, m_3 = 4 kg, m_4 = 1 kg, I_{x1} = 0.01 kg m^2, I_{y1} = 0.02 kg m^2, I_{z1} = 0.01 kg m^2, I_{x2} = 0.06 kg m^2, I_{y2} = 0.01 kg m^2, I_{z2} = 0.05 kg m^2, I_{x3} = 0.4 kg m^2, I_{y3} = 0.01 kg m^2, I_{z3} = 0.4 kg m^2, I_{x4} = 0.0005 kg m^2, I_{y4} = 0.001 kg m^2, and I_{z4} = 0.001 kg m^2.

Find and solve the equations of motion of the system.



C_4
4
L_3
q_3
3
C_3
2
q_4
1
A
C_2
q_2
L_2
C_1
q_1
L_1
0
$\mathbf{j}_0$
(0)
$\mathbf{\imath}_0$
$\mathbf{k}_0$
4
$\mathbf{j}_4$
q_3
(4)
$\mathbf{\imath}_4$
$\mathbf{k}_4$
$\mathbf{j}_2 \| \mathbf{j}_4$
3
$\mathbf{j}_2$
(2)
1
q_2
$\mathbf{\imath}_2$
$\mathbf{k}_2$
$\mathbf{j}_1$
$\mathbf{\imath}_1 \| \mathbf{\imath}_2$
2
q_4
q_1
(1)
$\mathbf{\imath}_1$
$\mathbf{k}_1$
Schematic representation
of the robot in 3D
$\mathbf{j}_0$
(0)
$\mathbf{k}_0$
$\mathbf{\imath}_0$

Figure 1

```
Off[General::spell]
Off[General::spell1]

(*Cross [ xx_ , yy_ ] :=
{ xx[[2]] yy[[3]]-xx[[3]] yy[[2]] ,
  xx[[3]] yy[[1]]-xx[[1]] yy[[3]] ,
  xx[[1]] yy[[2]]-xx[[2]] yy[[1]] };*)

(* Kinematics *)

(*transformation matrix from RF2 to RF1*)
R21= {{1,0,0},
      {0,  Cos[q2[t]], Sin[q2[t]]},
      {0, -Sin[q2[t]], Cos[q2[t]]}};

(*transformation matrix from RF4 to RF2*)
R42= {{Cos[q3[t]], 0, -Sin[q3[t]]},
      {0, 1, 0},
      {Sin[q3[t]], 0, Cos[q3[t]]} };

(* rule for expressing q1',q2',q3',q4' in terms
of generalized speeds u1,u2,u3,u4*)
rul ={
q1'[t]->(u1[t] Sin[q3[t]]-
         u3[t] Cos[q3[t]])/Sin[q2[t]],
q2'[t]->(u1[t] Cos[q3[t]]+u3[t] Sin[q3[t]]),
q3'[t]->u2[t]+(u3[t] Cos[q3[t]]-
        u1[t] Sin[q3[t]])*Cos[q2[t]]/Sin[q2[t]],
q4'[t]->u4[t]};

(*Angular velocities*)

(*Angular velocities of each link 1, 2, 3, 4, in
RF0, involving the generalized speeds, are expressed
using a vector basis fixed in the body
under consideration*)

(*anglar velocity of link 1 in RF0 expressed in
terms of RF1{i1,j1,k1}*)
w101={0,D[q1[t],t],0} /.rul ;

(*anglar velocity of link 1 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
w102 = w101.Transpose[R21] /.rul ;

(*anglar velocity of link 2 in RF1 expressed in
terms of RF1{i1,j1,k1}*)
w211={D[q2[t],t],0,0} /.rul ;


(*anglar velocity of link 2 in RF1 expressed in
terms of RF2{i2,j2,k2}*)
w212={D[q2[t],t],0,0}.Transpose[R21] /.rul ;

(*anglar velocity of link 2 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
w202=w102+w212;

(*anglar velocity of link 2 in RF0 expressed in
terms of RF4{i4,j2,k2}*)
w204=w202.Transpose[R42];
```

```
(*anglar velocity of link 3 in RF0 expressed in
terms of RF2{i2,j2,k2}==RF3{i3,j3,k3} *)
w302=w202;

(*anglar velocity of link 4 in RF2 expressed in
terms of RF2{i2,j2,k2}*)
w422= {0,D[q3[t],t],0} /.rul ;

(*anglar velocity of link 4 in RF2 expressed in
terms of RF4{i4,j4,k4}*)
w424= w422.Transpose[R42];

(*anglar velocity of link 4 in RF0 expressed in
terms of RF4{i4,j4,k4}*)
w404=Simplify[w424+w204 /.rul];

(*anglar velocity of link 4 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
w402= w404.R42;

(*angular accelerations*)

(*anglar acceleration of link 1 in RF0 expressed in
terms of RF1 {i1,j1,k1}*)
a101= D[w101,t] /.rul ;

(*anglar acceleration of link 2 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
a202=D[w202,t] /.rul ;

(*anglar acceleration of link 3 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
a302=a202;
```

```
(*anglar acceleration of link 4 in RF0 expressed in
terms of RF4{i4,j4,k4}*)
a404= D[w404,t];

(*
Remark: The angular velocity
w101 was expressed in terms of i1,j1,k1,
w202 was expressed in terms of i2,j2,k2,
w302 was expressed in terms of i2=i3,j2=j3,k2=k3,
w404 was expressed in terms of i4,j4,k4.
The reason for doing this is that it leads
automatically to expressions for
 D[w101,ur[t]] (r=1,2,3,4) in terms of i1,j1,k1,
 D[w202,ur[t]] (r=1,2,3,4) in terms of i2,j2,k2,
 D[w302,ur[t]] (r=1,2,3,4) in terms of i2,j2,k2,
 D[w404,ur[t]] (r=1,2,3,4) in terms of i4,j4,k4,
and this will facilitate later work, where it will be
assumed that the central principal axes of inertia of
link 1 are parallel to i1,j1,k1,
link 2 are parallel to i2,j2,k2,
link 3 are parallel to i2,j2,k2,
link 4 are parallel to i4,j4,k4.
```

```
When it comes to dealing with the velocities of
C1, C2, C3, C4, the mass centers of links 1, 2, 3, 4,
it is not necessarily advantageous to work with
i1,j1,k1, in the case of C1, with i2,j2,k2, for C2,
and so forth. Instead, it is best to use whatever
vector basis permits one to write the simplest
expression. *)

(*linear velocity of mass center C1 of link 1
in RF0 expressed in terms of RF1{i1,j1,k1} is
zero since C1 is fixed in RF0*)
vC101={0,0,0};

(*position vector from C1, mass center of link 1,
to C2, mass center of link 2, expressed in terms
of RF1{i1,j1,k1}*)
rC121={L1,L2,0};

(*linear velocity of mass center C2 of link 2 in
RF0 expressed in terms RF1{i1,j1,k1}*)
vC201=D[rC121,t]+Cross[w101,rC121] /.rul ;

(*Remark: velocity of C2, which is "fixed" in
RF1{i1,j1,k1}, can be computed as
vC201=Cross[w101,{L1,0,0}];*)




(*linear velocity of mass center C2 of link 2 in
RF0 expressed in terms RF2{i2,j2,k2}*)
vC202=vC201.Transpose[R21];

(*position vector from C1, mass center of link 1,
to C3, mass center of link 3, expressed in terms
of RF2{i2,j2,k2}*)
rC132={L1,L2,0}.Transpose[R21]+{0,q4[t],0};

(*linear velocity of mass center C3 of link 3 in
RF0 expressed in terms RF2{i2,j2,k2}*)
vC302=D[rC132,t]+Cross[w202,rC132] /.rul ;

(*Remark: another way of computing vC302
vC302=vC202+
D[{0,q4[t],0},t]+Cross[w202,{0,q4[t],0}]/.rul ;
*)

(*linear velocity of point C32 of link 2
expressed in terms of RF2{i2,j2,k2}
C32, of link 2, is superposed with C3, of link 3*)
vC3202=vC202+Cross[w202,{0,q4[t],0}];

(*linear velocity of mass center C4 of link 4 in
RF0 expressed in terms RF2 {i2,j2,k2}*)
vC402=vC302+Cross[w202,{0,L3,0}] /.rul ;

(*Linear accelerations*)

(*linear acceleration of mass center C1 of link 1 in
RF0 expressed in terms of RF1{i1,j1,k1}*)
aC101={0,0,0};

(*linear acceleration of mass center C2 of link 2 in
```

```
RF0 expressed in terms of RF1{i1,j1,k1}*)
aC201=D[vC201,t]+Cross[w101,vC201]/.rul ;

(*linear acceleration of mass center C3 of link 3 in
RF0 expressed in terms of RF2{i2,j2,k2}*)
aC302= D[vC302,t]+Cross[w202,vC302] /.rul ;

(*linear acceleration of mass center C4 of link 4 in
RF0 expressed in terms of RF2{i2,j2,k2}*)
aC402=D[vC402,t]+Cross[w402,vC402] /.rul ;

(*Dyadics*)

(*dyadic I1 of link 1 expressed in terms of
RF1{i1,j1,k1}*)
I1={{Ix1,0,0}, {0,Iy1,0},{0,0,Iz1}};

(*dyadic I2 of link 2 expressed in terms of
RF2{i2,j2,k2}*)
I2={{Ix2,0,0}, {0,Iy2,0},{0,0,Iz2}};

(*dyadic I3 of link 3 expressed in terms of
RF2{i2,j2,k2}*)
I3={{Ix3,0,0}, {0,Iy3,0},{0,0,Iz3}};

(*dyadic I4 of link 4 expressed in terms of
RF4{i4,j4,k4}*)
I4={{Ix4,0,0}, {0,Iy4,0},{0,0,Iz4}};

(*Inertia torques*)

(*inertia torque Tin1 of link 1 in RF0 expressed in
terms of RF1{i1,j1,k1} *)
Tin1= -a101.I1-Cross[w101,I1.w101];

(*inertia torque Tin2 of link 2 in RF0 expressed in
terms of RF1{i2,j2,k2}*)
Tin2= -a202.I2-Cross[w202,I2.w202];

(*inertia torque Tin3 of link 3 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
Tin3= -a302.I3-Cross[w202,I3.w302];

(*inertia torque Tin4 of link 4 in RF0 expressed in
terms of RF4{i4,j4,k4} *)
Tin4=-a404.I4-Cross[w404,I4.w404];

(*Inertia force*)

(*inertia force Fin1 of link1 in RF0 expressed in
terms of RF1{i1,j1,k1}*)
Fin1=-m1 aC101;

(*inertia force Fin2 of link2 in RF0 expressed in
terms of RF1{i1,j1,k1}*)
Fin2=-m2 aC201;

(*inertia force Fin3 of link3 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
Fin3=-m3 aC302;

(*inertia force Fin4 of link4 in RF0 expressed in
terms of RF2{i2,j2,k2}*)
Fin4=-m4 aC402;
```

```
(*Generalized inertia forces*)

Kin1=
D[Expand[w101 ],u1[t]].Tin1+
D[Expand[vC101],u1[t]].Fin1+
D[Expand[w202 ],u1[t]].Tin2+
D[Expand[vC201],u1[t]].Fin2+
D[Expand[w302 ],u1[t]].Tin3+
D[Expand[vC302],u1[t]].Fin3+
D[Expand[w404 ],u1[t]].Tin4+
D[Expand[vC402],u1[t]].Fin4;

Kin2=
D[Expand[w101 ],u2[t]].Tin1+
D[Expand[vC101],u2[t]].Fin1+
D[Expand[w202 ],u2[t]].Tin2+
D[Expand[vC201],u2[t]].Fin2+
D[Expand[w302 ],u2[t]].Tin3+
D[Expand[vC302],u2[t]].Fin3+
D[Expand[w404 ],u2[t]].Tin4+
D[Expand[vC402],u2[t]].Fin4;

Kin3=
D[Expand[w101 ],u3[t]].Tin1+
D[Expand[vC101],u3[t]].Fin1+
D[Expand[w202 ],u3[t]].Tin2+
D[Expand[vC201],u3[t]].Fin2+
D[Expand[w302 ],u3[t]].Tin3+
D[Expand[vC302],u3[t]].Fin3+
D[Expand[w404 ],u3[t]].Tin4+
D[Expand[vC402],u3[t]].Fin4;



Kin4=
D[Expand[w101 ],u4[t]].Tin1+
D[Expand[vC101],u4[t]].Fin1+
D[Expand[w202 ],u4[t]].Tin2+
D[Expand[vC201],u4[t]].Fin2+
D[Expand[w302 ],u4[t]].Tin3+
D[Expand[vC302],u4[t]].Fin3+
D[Expand[w404 ],u4[t]].Tin4+
D[Expand[vC402],u4[t]].Fin4;


(*Contact and gravitational force*)

(*rigid link 1*)

(*force applied by base 0 to link 1 at C1
expressed in terms of RF1{i1,j1,k1}*)
K01={K01x, K01y, K01z};

(*torque applied by base 0 to link 1
expressed in terms of RF1{i1,j1,k1}*)
T01={T01x, T01y, T01z};

(*force applied by link 2 to link 1 at C2
expressed in terms of RF2{i2,j2,k2}*)
K21={K21x, K21y, K21z};

(*torque applied by link 2 to link 1
```

```
expressed in terms of RF2{i2,j2,k2}*)
T21={T21x, T21y, T21z};

(*gravitational force that acts on link 1 at C1
expressed in terms of RF1{i1,j1,k1}*)
G1={0,-m1 g, 0};

(*generalized active forces for link 1*)

K1a1=
D[Expand[w101 ],u1[t]].T01+
D[Expand[vC101],u1[t]].K01+
D[Expand[w101 ],u1[t]].Transpose[R21].(T21)+
D[Expand[vC201],u1[t]].Transpose[R21].(K21)+
D[Expand[vC101],u1[t]].G1;

K1a2=
D[Expand[w101 ],u2[t]].T01+
D[Expand[vC101],u2[t]].K01+
D[Expand[w101 ],u2[t]].Transpose[R21].(T21)+
D[Expand[vC201],u2[t]].Transpose[R21].(K21)+
D[Expand[vC101],u2[t]].G1;


K1a3=
D[Expand[w101 ],u3[t]].T01+
D[Expand[vC101],u3[t]].K01+
D[Expand[w101 ],u3[t]].Transpose[R21].(T21)+
D[Expand[vC201],u3[t]].Transpose[R21].(K21)+
D[Expand[vC101],u3[t]].G1;

K1a4=
D[Expand[w101 ],u4[t]].T01+
D[Expand[vC101],u4[t]].K01+
D[Expand[w101 ],u4[t]].Transpose[R21].(T21)+
D[Expand[vC201],u4[t]].Transpose[R21].(K21)+
D[Expand[vC101],u4[t]].G1;

(*rigid link 2*)

(*force applied by link 1 to link 2 at C2
expressed in terms of RF2{i2,j2,k2}: -K21 *)

(*torque that applied by link 1 to link 2
expressed in terms of RF2{i2,j2,k2}: -T21 *)

(*force applied by link 3 to link 2 at C32
expressed in terms of RF2{i2,j2,k2}*)
K32={K32x, K32y, K32z};

(*torque applied by link 3 to link 2
expressed in terms of RF2{i2,j2,k2}*)
T32={T32x, T32y, T32z};

(*gravitational force that acts on link 2 at C2
expressed in terms of RF2{i2,j2,k2}*)
G2={0,-m2 g, 0};

(*generalized active forces for link 2*)

K2a1=
D[Expand[w202  ],u1[t]].(-T21)+
D[Expand[vC201 ],u1[t]].Transpose[R21].(-K21)+
D[Expand[w202  ],u1[t]].T32+
```

```
D[Expand[vC3202],u1[t]].K32+
D[Expand[vC201 ],u1[t]].G2;

K2a2=
D[Expand[w202  ],u2[t]].(-T21)+
D[Expand[vC201 ],u2[t]].Transpose[R21].(-K21)+
D[Expand[w202  ],u2[t]].T32+
D[Expand[vC3202],u2[t]].K32+
D[Expand[vC201 ],u2[t]].G2;

K2a3=
D[Expand[w202  ],u3[t]].(-T21)+
D[Expand[vC201 ],u3[t]].Transpose[R21].(-K21)+
D[Expand[w202  ],u3[t]].T32+
D[Expand[vC3202],u3[t]].K32+
D[Expand[vC201 ],u3[t]].G2;

K2a4=
D[Expand[w202  ],u4[t]].(-T21)+
D[Expand[vC201 ],u4[t]].Transpose[R21].(-K21)+
D[Expand[w202  ],u4[t]].T32+
D[Expand[vC3202],u4[t]].K32+
D[Expand[vC201 ],u4[t]].G2;

(*rigid link 3*)

(*force applied by link 2 to link 3 at C3
expressed in terms of RF2{i2,j2,k2}: -K32 *)

(*torque applied by link 2 to link 3
expressed in terms of RF2{i2,j2,k2}: -T32 *)

(*force that applied by link 4 to link 3 at C4
expressed in terms of RF2{i2,j2,k2}*)
K43={K43x, K43y, K43z};

(*torque that applied by link 4 to link 3
expressed in terms of RF2 {i2,j2,k2}*)
T43={T43x, T43y, T43z};

(*gravitational force that acts on link 3 at C3
expressed in terms of RF2{i2,j2,k2}*)
G3={0,-m3 g, 0}.Transpose[R21];

(*generalized active forces for link 3*)

K3a1=
D[Expand[w302 ],u1[t]].(-T32)+
D[Expand[vC302],u1[t]].(-K32)+
D[Expand[w302 ],u1[t]].T43+
D[Expand[vC402],u1[t]].K43+
D[Expand[vC302],u1[t]].G3;

K3a2=
D[Expand[w302 ],u2[t]].(-T32)+
D[Expand[vC302],u2[t]].(-K32)+
D[Expand[w302 ],u2[t]].T43+
D[Expand[vC402],u2[t]].K43+
D[Expand[vC302],u2[t]].G3;
```

```
K3a3=
D[Expand[w302 ],u3[t]].(-T32)+
D[Expand[vC302],u3[t]].(-K32)+
D[Expand[w302 ],u3[t]].T43+
D[Expand[vC402],u3[t]].K43+
D[Expand[vC302],u3[t]].G3;

K3a4=
D[Expand[w302 ],u4[t]].(-T32)+
D[Expand[vC302],u4[t]].(-K32)+
D[Expand[w302 ],u4[t]].T43+
D[Expand[vC402],u4[t]].K43+
D[Expand[vC302],u4[t]].G3;


(*rigid link 4*)

(*force that applied by link 3 to link 4 at C4
expressed in terms of RF2{i2,j2,k2}: -K43 *)

(*torque that applied by link 4 to link 3
expressed in terms of RF2 {i2,j2,k2}: -T43*)

(*gravitational force that acts on link 4 at C4
expressed in terms of RF2{i2,j2,k2}*)
G4= {0, -m4 g, 0}.Transpose[R21];

(*generalized active forces for link 4*)

K4a1=
D[Expand[w404 ],u1[t]].R42.(-T43)+
D[Expand[vC402],u1[t]].(-K43)+
D[Expand[vC402],u1[t]].G4;

K4a2=
D[Expand[w404 ],u2[t]].R42.(-T43)+
D[Expand[vC402],u2[t]].(-K43)+
D[Expand[vC402],u2[t]].G4;

K4a3=
D[Expand[w404 ],u3[t]].R42.(-T43)+
D[Expand[vC402],u3[t]].(-K43)+
D[Expand[vC402],u3[t]].G4;

K4a4=
D[Expand[w404 ],u4[t]].R42.(-T43)+
D[Expand[vC402],u4[t]].(-K43)+
D[Expand[vC402],u4[t]].G4;

(*Generalized active forces *)

K1=Simplify[Expand[K1a1+K2a1+K3a1+K4a1]];
K2=Simplify[Expand[K1a2+K2a2+K3a2+K4a2]];
K3=Simplify[Expand[K1a3+K2a3+K3a3+K4a3]];
K4=Simplify[Expand[K1a4+K2a4+K3a4+K4a4]];

(*input numerical data*)
indata={
L1->0.1, L2->0.1, L3->0.7,
m1->9., m2->6., m3->4., m4->1.,
```

```
Ix1->0.01, Iy1->0.02, Iz1->0.01,
Ix2->0.06, Iy2->0.01, Iz2->0.05,
Ix3->0.4, Iy3->0.01, Iz3->0.4,
Ix4->0.0005, Iy4->0.001, Iz4->0.001,
k1->3.,k2->5.,k3->1.,k4->3.,k5->0.3,
k6->0.6,k7->30.,k8->41.,
g->9.8,
p1->N[Pi/3],p2->N[Pi/3],p3->N[Pi/3],p4->0.1};

(*control torques and force*)
control ={
T01y->k1 (p1-q1[t])- k2  q1'[t],
T21x->k3 (q2[t]-p2)+k4  q2'[t]+
      g ((m3+m4) q4[t]+m4 L3) Sin[q2[t]],
T43y->k5 (q3[t]-p3)+k6  q3'[t],
K32y->k7 (q4[t]-p4)+k8  q4'[t]-g (m3+m4) Cos[q2[t]]
          } /.indata/.rul ;

(* Kane's dynamical equations *)
e1=K1+Kin1 /.indata /.control;
e2=K2+Kin2 /.indata /.control;
e3=K3+Kin3 /.indata /.control;
e4=K4+Kin4 /.indata /.control;
eqs ={
(*kinematic equations*)
u1[t]== q1'[t] Sin[q2[t]] Sin[q3[t]] +
        q2'[t] Cos[q3[t]],
u2[t]== q1'[t] Cos[q2[t]]+q3'[t],
u3[t]==-q1'[t] Cos[q3[t]] Sin[q2[t]] +
        q2'[t] Sin[q3[t]],
u4[t]== q4'[t],
(*dynamical equations*)
e1==0, e2==0, e3==0, e4==0,
(*initial conditions*)
q1[0]==N[Pi/6],
q2[0]==N[Pi/12],
q3[0]==N[Pi/10],
q4[0]==0.01,
u1[0]==0,u2[0]==0,u3[0]==0,u4[0]==0 };

(*Numerical simulation of the equations of motion*)
kane = NDSolve [eqs,{q1,q2,q3,q4,u1,u2,u3,u4},
       {t,0,15}]
Plot [Evaluate[q1[t] /. kane], {t, 0, 15},
PlotRange->{All,All},AxesLabel->{ "t[s]","q1[rad]"}];
Plot [Evaluate[q2[t] /. kane], {t, 0, 15},
PlotRange->{All,All},AxesLabel->{ "t[s]","q2[rad]"}];
Plot [Evaluate[q3[t] /. kane], {t, 0, 15},
PlotRange->{All,All},AxesLabel->{ "t[s]","q3[rad]"}];
Plot [Evaluate[q4[t] /. kane], {t, 0, 15},
PlotRange->{All,All},AxesLabel->{ "t[s]","q4[m]"}];
```

```
Out[186]= {{q1 Æ InterpolatingFunction[{{0., 15.}}, <>],
  q2 Æ InterpolatingFunction[{{0., 15.}}, <>],
  q3 Æ InterpolatingFunction[{{0., 15.}}, <>],
  q4 Æ InterpolatingFunction[{{0., 15.}}, <>],
  u1 Æ InterpolatingFunction[{{0., 15.}}, <>],
  u2 Æ InterpolatingFunction[{{0., 15.}}, <>],
  u3 Æ InterpolatingFunction[{{0., 15.}}, <>],
  u4 Æ InterpolatingFunction[{{0., 15.}}, <>]}}
```


q1[rad]
t[s]
2
4
6
8
10
12
14
0.9
0.8
0.7
0.6


q2[rad]
t[s]
2
4
6
8
10
12
14
0.8
0.6
0.4


q3[rad]
t[s]
2
4
6
8
10
12
14
0.9
0.8
0.7
0.6
0.5
0.4
0.3


q4[m]
t[s]
2
4
6
8
10
12
14
0.08
0.06
0.04
0.02